

## Основные типы

```
integer real boolean char string BigInteger
```

### Пример программы 1

```
// Площадь и периметр прямоугольника
// Особенности: функция ввода ReadReal, описание с инициализацией и выводом типа
begin
  var a := ReadReal('Введите a:'); // внутриблочное описание
  var b := ReadReal('Введите b:'); // с инициализацией вводимым значением
  var (S,P) := (a*b,2*(a+b)); // множественная инициализация
  Println('Площадь =', S); // Println разделяет данные пробелами
  Println('Периметр =', P);
end.
```

### Пример программы 2

```
{ Факториал числа. Расширенные операторы присваивания,
  BigInteger, for var i, интерполированные строки }
begin
  var n := ReadInteger;
  var p: BigInteger := 1; // тип длинного целого
  for var i:=1 to n do // описание счётчика цикла for в заголовке
    p *= i; // расширенный оператор присваивания *=
  Println('${n}!={p}'); // вывод интерполированной строки
end.
```

### Расширенные операторы присваивания

```
x += 1; // x := x + 1
a *= 2; // a := a * 2   Есть -= и /=
```

### Print и Write

```
Write(1,2,3); // 123
Print(1,2,3); // 1 2 3
```

### Инициализация при описании, автовывод типа

```
var счётчик: integer := 0;
var b := 2.5; // автовывод: тип real
var s := 'Строка'; // автовывод: тип string
```

### Множественное присваивание, множественная инициализация

```
var (a,b) := (3,5);
(a,b) := (a+1,b*2);
(a,b) := (b,a); // меняет местами a и b
(a,b,c) := (b,c,a); // циклически влево сдвигает значения в a,b,c
```

### Функции ввода

```
var n := ReadInteger;
var x := ReadReal('Введите x:');
var (a,b) := ReadReal2('Введите a,b:');
var b: boolean := TryRead(x);
```

### Полезные константы, функции, методы, операции

```
integer.MinValue integer.MaxValue i.InRange(3,5) i.IsEven i.IsOdd
real.MinValue real.MaxValue r.InRange(3,5) 2**10 3**0.5
Max(a,b) Min(a,b) Round(x) x.Round
Swap(x,y) // меняет местами значения x и y
```

## Управляющие операторы

### Условный оператор

```
var (a,b) := (1,1);  
if (a,b) = (1,1) then  
    (a,b) := (a+1,b-1);
```

```
if a<b then  
    min := a  
else min := b;
```

```
var (a,b) := (5,3);  
if a>b then  
begin  
    var c := a;  
    a := b;  
    b := c;  
end;
```

### Оператор выбора

```
case Месяц of  
    12,1,2: Сезон := 'зима';  
    3..5: Сезон := 'весна';  
    6..8: Сезон := 'лето';  
    9..11: Сезон := 'осень';  
else Сезон := 'ошибка';  
end;
```

### Циклы

```
for var i:=1 to n do  
    Print(i);
```

```
loop n do  
    Print(1);
```

```
var x := 1.0;  
repeat  
    x *= 2  
until x>1000;
```

```
while x>1 do  
    x /= 2;
```

```
var сумма := 0;  
foreach var x in ReadSeqInteger(10) do  
    сумма += x;  
// ReadSeqInteger(10) - последовательность n целых, вводимых с клавиатуры
```